

Math 216 Differential Equations

Lab 4: Solving Differential Equations Using Maple

Goals

In this lab you will learn how to use Maple to solve differential equations analytically (when that is possible) or numerically; then you will learn how to plot the solutions. In the prelab you will first learn how to write differential equations in Maple notation.

A Maple primer

We start with a quick review of Maple syntax as we need it to solve differential equations. If you are already comfortable with Maple you can skip directly to the prelab assignment. You need to be comfortable with the aspects of Maple described in this primer *before* you go to lab.

Writing differential equations in Maple notation

In what appears below, the input to Maple is typed after the prompt `>` and the output from Maple is indented.

The derivative of an expression $y(t)$ with respect to t is denoted in Maple by `diff(y(t),t)`. To enter the differential equation $ty' + 5y = \ln(t)$ into Maple we type the line below and hit the return key.

```
> t*diff(y(t),t)+5*y(t)=ln(t);
```

$$t \left(\frac{d}{dt} y(t) \right) + 5y(t) = \ln(t)$$

The output produced by Maple is a “pretty print” of the equation. Note that we refer to y consistently as $y(t)$ in communicating with Maple.

It is useful to give the equation we’ve just entered a name, and also to add some spaces for readability. When you give something a name in Maple you use the assignment operator `:=`, that is, a colon `:` followed by an equal sign `=` (with no spaces in between). In writing an equation however you use just the equal sign `=`.

Let’s go back and change the entry to

```
> ex1 := t*diff(y(t),t) + 5*y(t) = ln(t);
```

$$\text{ex1} := t \left(\frac{d}{dt} y(t) \right) + 5y(t) = \ln(t)$$

Note that you must end every entry with a semicolon or colon if you want Maple to pay attention. Use a colon only if you do *not* want to see the result; this can be useful if you expect the result to be a very long expression. Make sure to hit the return key after each command that you want Maple to execute.

The second derivative of an expression $y(t)$ with respect to t is denoted in communicating with Maple by `diff(y(t),t,t)`. Here is how you enter the differential equation $y'' + 4y = t$:

```
> ex2 := diff(y(t),t,t) + 4*y(t) = t;
```

$$\text{ex2} := \left(\frac{d^2}{dt^2} y(t) \right) + 4y(t) = t$$

You can guess how to write the third derivative.

Again, when you give something a name in Maple you use the assignment operator `:=` as in the following example.

```
> b := x^2 + 3*sin(x) - 7;
```

$$b := x^2 + 3 \sin(x) - 7$$


Since `b` has been defined as the name of an expression, we can differentiate it, or plot it, or evaluate it at $x = 3$, and so on:

```
> diff(b,x);
```

$$2x + 3 \cos(x)$$


This is the command for plotting an expression:

```
> plot(b, x = -Pi..2);
```



This is how you plot two functions on the same graph:

```
> plot({b,diff(b,x)}, x =-Pi..2);
```



This is how you evaluate the expression at a point:

```
> eval(b, x = 3);
```

$2 + 3 \sin(3)$

This is how you approximate your answer by a decimal. The ditto operator % refers to the last result output by Maple.

```
> evalf(%);
```

2.423360024

Now you should be able to write down equations and have Maple understand them. Some standard functions that are known to Maple may be helpful: **abs(x)** means $|x|$, **exp(x)** means e^x , and **sin(x)**, **arcsin(x)**, **arctan(x)**, **sec(x)**, **cot(x)**, **ln(x)**, and **csc(x)** mean exactly what you think they do.

A final note. It is easy to get help on how to use a command that Maple knows by typing **?** followed by the command name and hitting return (no need for a semicolon or colon here). A window then appears containing information about how to use the command. What happens if you don't exactly know the name of the command you should use? Frequently Maple is smart enough to give you information even if you don't know the precise name for the command you want to ask about. For example, there is no command in Maple called **cotangent**, but if you type

```
> ?cotangent
```

and hit return, you get a window containing the mathematical definition of the cotangent function, and a link called [trig](#). Following this link leads to a page containing the Maple syntax for the `cot(x)` function.

In summary, when you are typing commands into Maple,

- Always end commands with a `;` or `:` and by pressing the return key.
- Always use `*` to indicate multiplication.
- Use `:=` for assignment and reserve `=` for typing equations.
- Use `y(t)` or `y(x)` in your equations when you are referring to a function, not just `y`.
- Use `diff(y(x),x)` to indicate a derivative, not `y'(x)`.

Solving differential equations analytically

Now that you know how to enter equations into Maple we look at some examples using Maple to solve differential equations analytically, that is, expressing the answer in terms of known functions. Pay attention to the syntax; your turn is coming soon.

We will first ask for the general solution of the differential equation $y'' + y = t^5 + \cos(3t)$. First, we define the equation by giving it a name:

```
> ex3 := diff(y(t),t,t) + y(t) = t^5 + cos(3*t);
```

$$\text{ex3} := \left(\frac{d^2}{dt^2} y(t) \right) + y(t) = t^5 + \cos(3t)$$

The output is a “pretty print” of the equation. Now we ask Maple to solve the differential equation for the unknown function $y(t)$:

```
> dsolve(ex3,y(t));
```

$$y(t) = \sin(t) _C2 + \cos(t) _C1 + t^5 - 20t^3 + 120t - \frac{1}{8} \cos(3t)$$

Note that Maple uses `_C1`, `_C2`, and so on as constants of integration. It might be nicer if Maple wrote `_C2 sin(t)` rather than `sin(t) _C2` but unfortunately it doesn't. However, you should also notice that the command syntax isn't very complicated. Indeed, the Maple command `dsolve(equation,function)` means “solve the differential equation `equation` for the unknown `function`”.

We will now ask Maple for the particular solution of the differential equation $y'' + y = t^5 + \cos(3t)$ satisfying the initial conditions $y(0) = 1/3$ and $y'(0) = 2/5$.

```
dsolve({ex3, y(0) = 1/3, D(y)(0) = 2/5},y(t));
```

$$y(t) = -\frac{598}{5} \sin(t) + \frac{11}{24} \cos(t) + t^5 - 20t^3 + 120t - \frac{1}{8} \cos(3t)$$

Note that in this situation, **D** stands for derivative so you enter the initial condition $y'(0) = 2/5$ by specifying the value of the derivative of y , **D(y)**, evaluated at $t = 0$: **D(y)(0)**. Now you can see that the more general form of the **dsolve** command that is appropriate for solving initial-value problems is therefore **dsolve({equation,initial_conditions},function)**. Beware: in Maple, wiggly brackets are not the same as parentheses.

The solution Maple has provided to the initial-value problem is free of arbitrary constants so we can plot it. The result of using **dsolve** was an equation so we can give a name to the right-hand side as follows:

```
> Y1:=rhs(%);
```

$$Y1 := -\frac{598}{5} \sin(t) + \frac{11}{24} \cos(t) + t^5 - 20t^3 + 120t - \frac{1}{8} \cos(3t)$$


Recall here that the ditto operator **%** always refers to the previous result. Now that we have named the solution we can evaluate it. If we want the value of the solution at $t = 0.32$ we enter

```
> eval(Y1,t = .32);
```

.4892110217

Plots of solutions are often useful. To plot the solution we enter

```
> plot(Y1,t = -1..1);
```



This plots the solution for $-1 \leq t \leq 1$; the notation **t = -1..1** indicates this range of t -values.

Solving differential equations numerically

Exact analytical solutions to differential equations can sometimes not be found. Maple does not always succeed when instructed to find the solution using **dsolve**. For example, here

is a simple second-order linear homogeneous ode: $y'' + y' + x^3y = 0$. Suppose we enter the equation into Maple and name it:

```
> ex4:= diff(y(x),x,x)+diff(y(x),x)+x^3*y(x)=0;
```

$$\text{ex4} := \left(\frac{d^2}{dx^2} y(x) \right) + \left(\frac{d}{dx} y(x) \right) + x^3 y(x) = 0$$

Now we ask Maple to find the general solution of this differential equation:

```
> dsolve(ex4,y(x));
```

$$y(x) = \text{DESol} \left(\left\{ x^3 -Y(x) + \left(\frac{d}{dx} -Y(x) \right) + \left(\frac{d^2}{dx^2} -Y(x) \right) \right\}, \{-Y(x)\} \right)$$

As you can see, Maple cannot solve this and eventually responds just by recopying the problem.

Most differential equations do not have explicit solutions in terms of known functions; the best you can do is to find numerical solutions. In Lab 2 and Lab 3 we learned how to do this using Matlab. However, the numerical schemes we implemented ourselves in Matlab are already built-in as part of Maple. Here is how to use Maple to solve initial-value problems for differential equations numerically.

In order to solve the differential equation numerically, we need to have numerical initial conditions instead of symbolic constants. For example, if $y(0) = 3$ and $y'(0) = -2$ we can solve the initial-value problem numerically, but if $y(0) = c$ and $y'(0) = n$, we cannot without saying what values c and n take. To tell Maple to solve the initial-value problem numerically, just type

```
> ans4:=dsolve({ex4,y(0)=3,D(y)(0)=-2},y(x),numeric);
```

```
ans4 := proc(x_rkf45) ... end proc;
```

The word **numeric** tells Maple to calculate the solution numerically rather than analytically. Maple returns a procedure (a type of function) as its answer. To get values from that procedure we enter:

```
> ans4(.3);
```

$$\left[x = 0.3, y(x) = 2.48133128459059904, \frac{d}{dx} y(x) = -1.48654476849350292 \right]$$

Now we know the values of y and y' at $x = 0.3$.

In addition to getting the value of the solution at a point we may wish to plot the solution. The easiest way to do this is to first load a package called **plots**. It is only necessary to do this once in each Maple session:


```
> with(plots):
```

```
Warning, the name changecoords has been redefined
```

You can safely ignore the warning that appears here. That is how packages are loaded into Maple. Here we used a colon (:) after the command rather than a semicolon (;) to prevent the Maple worksheet from filling up with a list of all the new commands available in the package `plots`.


Having loaded the package `plots` we can use the `odeplot` command to plot y versus x , that is, x on the horizontal axis and y on the vertical axis, as follows:

```
> odeplot(ans4,[x, y(x)], 0..1);
```




To plot y' versus x , that is, x on the horizontal axis and y' on the vertical axis, type

```
> odeplot(ans4,[x, diff(y(x),x)], 0..1);
```



You can also plot both y and y' versus x , by using an extra set of brackets (`[]`) in this case:

```
> odeplot(ans4, [[x, y(x)], [x, diff(y(x),x)]], 0..1);
```



Prelab assignment

Before arriving in the lab, answer the following questions. You will need these results to finish the lab problems, and your recitation instructor may check that you have brought them. These problems are to be handed in as part of your lab report.

1. For each part of this problem, write down the Maple commands to solve the given equation. The answer to part 1a is given below; you must do the others.

(a) $x^2y' + xy - y^2 = 0$. Answer:

```
> eq1a:=x^2*diff(y(x),x)+x*y(x)-y(x)^2=0; dsolve(eq1a,y(x));
```

This answer shows that it is okay to put two consecutive commands on the same line; you only have to hit return once.

(b) $y' = \frac{xy + y^2}{x^2}$.

(c) $y'' + y' + y = te^{-t}$. Note that here the independent variable is t , not x .

(d) $y'' + 25y = \cos(t)^2$. Note that here also the independent variable is t .

2. In each part of this problem write down the Maple commands to solve the given initial-value problem. The answer to part 2a is given below; you must do the others.

(a) $x^2y' + xy - y^2 = 0$, subject to $y(1) = 1/3$. Answer:

```
> dsolve({eq1a,y(1)=1/3},y(x));
```

(b) $y' = \frac{xy + y^2}{x^2}$, subject to $y(1) = 2/3$.

(c) $y'' + y' + y = te^{-t}$, subject to $y(0) = 0$ and $y'(0) = -1$.

(d) $y'' + 25y = \cos(t)^2$, subject to $y(0) = 0$ and $y'(0) = 0$.

3. In each part of this problem write down the Maple commands to solve the *system* of differential equations. The answer to part 3a is given below; you must do part 3b.

(a) $x' = -5x + 17y$, $y' = -2x + 5y + \sin(5t)$, $x(0) = 1$, $y(0) = -3$. Answer:

```
> eq3a1 := diff(x(t), t) = -5*x(t) + 17*y(t):  
> eq3a2 := diff(y(t), t) = -2*x(t) + 5*y(t) + sin(5*t):  
> dsolve({eq3a1,eq3a2,x(0)= 1,y(0)=-3}, {x(t),y(t)});
```

(b) $x' = -3x + 4y$, $y' = 5x - 7y + \cos(t)$, $x(0) = 0$, $y(0) = -1$.

In the lab

After logging on to the Macintosh system in the Instructional Computer Labs located in the basement of East Hall, launch Maple by clicking on the leaf-shaped icon in the dock. A window called **untitled** eventually appears, and you should see the Maple prompt **>** at the start of the first line. To execute the example commands given below, you should position the cursor after the Maple prompt **>** and then type the remaining part of the line (do not type **>** itself) and hit return.

If you have been working with Maple for some time and need to restart your work for some reason, it is important to tell Maple to “forget” all previous assignment statements to avoid possible confusion. The command for doing this is

```
> restart;
```

Before you start typing commands in the Maple window, give the worksheet the name `Lab4.mws` by clicking on the floppy disk icon or by choosing `Save` from the `File` menu (when doing this, make sure that you are saving the file in your home directory `~/Desktop/IFS`). Once you are working, save your work frequently. At the end of the lab, you can print your Maple worksheet by clicking on the printer icon or by choosing `Print...` from the `File` menu.

Lab problems

1. In each part of this problem refer to your prelab assignment answers. The solution to part 1a is given below; you must do the others.

- (a) Find $y(2)$ and plot $y(x)$ for $1 \leq x \leq 2$ if $y(x)$ is the solution of the initial-value problem $x^2y' + xy - y^2 = 0$, subject to $y(1) = 1/3$. Answer: from the prelab assignment we know how to define the differential equation for Maple:

```
> eq1a:= x^2*diff(y(x),x) + x*y(x) - y(x)^2 = 0;
```

$$\text{eq1a} := x^2 \left(\frac{d}{dx} y(x) \right) + x y(x) - y(x)^2 = 0$$

Also from the prelab assignment we know how to ask Maple to solve the specified initial-value problem for this differential equation. We add an assignment with the `:=` operator to give the solution a name:


```
> ans1a:=dsolve({eq1a,y(1)=1/3},y(x));
```

$$\text{ans1a} := y(x) = \frac{2x}{1 + 5x^2}$$

Having named the solution, we can evaluate it and plot it to finish the problem:

```
> Y1a:=rhs(ans1a); eval(Y1a,x=2); plot(Y1a,x=1..2);
```

$$Y1a := \frac{2x}{1 + 5x^2}$$
$$\frac{4}{21}$$



This concludes the solution of part 1a.

- (b) Find $y(2)$ and plot $y(x)$ for $1 \leq x \leq 2$ if $y(x)$ is the solution of the initial-value problem $y' = \frac{xy + y^2}{x^2}$, subject to $y(1) = 2/3$.
- (c) Find $y(2)$ and plot $y(t)$ for $0 \leq t \leq 10$ if $y(t)$ is the solution of the initial-value problem $y'' + y' + y = te^{-t}$, subject to $y(0) = 0$ and $y'(0) = -1$.
- (d) Find $y(2)$ and plot $y(t)$ for $0 \leq t \leq 20$ if $y(t)$ is the solution of the initial-value problem $y'' + 25y = \cos(t)^2$, subject to $y(0) = 0$, $y'(0) = 0$.
2. In each part of this problem use your prelab assignment answers to help you evaluate $x(20)$, plot x versus t for $0 \leq t \leq 20$, and then plot y versus x for $0 \leq t \leq 20$. The solution to part 2a is given below; you must complete part 2b.

- (a) $x' = -5x + 17y$, $y' = -2x + 5y + \sin(5t)$, $x(0) = 1$, $y(0) = -3$. Answer: begin by defining the two differential equations according to the results of the prelab assignment:

```
> eq2a1 := diff(x(t),t)=-5*x(t)+17*y(t);
```

$$\text{eq2a1} := \frac{d}{dt}x(t) = -5x(t) + 17y(t)$$

```
> eq2a2 := diff(y(t),t)=-2*x(t)+5*y(t)+sin(5*t);
```

$$\text{eq2a2} := \frac{d}{dt}y(t) = -2x(t) + 5y(t) + \sin(5t)$$

Next, solve the system of differential equations with the given initial conditions, again using a result from the prelab assignment:

```
> ans2a := dsolve({eq2a1,eq2a2,x(0)=1,y(0)=-3},{x(t),y(t)});
```

$$\text{ans2a} := \left\{ \begin{array}{l} x(t) = -\frac{811}{48}\sin(3t) + \cos(3t) - \frac{17}{16}\sin(5t), \\ y(t) = -\frac{43}{16}\cos(3t) - \frac{247}{48}\sin(3t) - \frac{5}{16}\cos(5t) - \frac{5}{16}\sin(5t) \end{array} \right\}$$

The answer here is a pair of equations. To extract the function $x(t)$ we refer to the first part of the answer as `ans2a[1]` and to extract the function $y(t)$ we refer to the second part of the answer as `ans2a[2]`. For example,

```
> ans2a[1];
```

$$x(t) = -\frac{811}{48}\sin(3t) + \cos(3t) - \frac{17}{16}\sin(5t)$$

Therefore to select and name the two functions we can type

```
> Xa:=rhs(ans2a[1]); Ya:=rhs(ans2a[2]);
```

$$Xa := -\frac{811}{48}\sin(3t) + \cos(3t) - \frac{17}{16}\sin(5t)$$

$$Ya := -\frac{43}{16}\cos(3t) - \frac{247}{48}\sin(3t) - \frac{5}{16}\cos(5t) - \frac{5}{16}\sin(5t)$$

To evaluate $x(t)$ at $t = 20$, we can type

```
> eval(Xa,t=20);
```

$$-\frac{811}{48}\sin(60) + \cos(60) - \frac{17}{16}\sin(100)$$


To see a floating point version of this result, we can type

```
> evalf(eval(Xa,t=20));
```

4.735629965


To plot $x(t)$ as a function of t , type

```
> plot(Xa,t=0..20);
```



Finally, to make a parametric plot of $y(t)$ versus $x(t)$, we use a modification of the `plot` command:

```
> plot([Xa,Ya,t=0..20]);
```



This completes the solution of part 2a.

(b) $x' = -3x + 4y$, $y' = 5x - 7y + \cos(t)$, $x(0) = 0$, $y(0) = -1$.

- The remaining problems involve using Maple to solve initial-value problems for differential equations numerically. Start by loading the `plots` package if you have not done this already. Find $y(1)$ and $y'(1)$, and then plot $y(x)$ and $y'(x)$ on the same graph over the range $0 \leq x \leq 1$, if $y(x)$ is the solution of the initial-value problem corresponding

to the differential equation $y'' + \sin(xy) = 0$ and the initial conditions $y(0) = 1$ and $y'(0) = 2$.

4. You can also solve systems numerically using Maple in the same way you solve a single equation. For each part of this problem find $x(1)$ and $y(1)$, and then plot $x(t)$ and $y(t)$ on the same graph over the range $0 \leq t \leq 2$, and finally plot $y(t)$ versus $x(t)$ for $0 \leq t \leq 2$. The solution to part 4a is given below. You must do part 4b.

- (a) $x' = -5x + 17y$, $y' = -2x^3 + 5y$, subject to $x(0) = 1$, $y(0) = -3$. Answer: start by defining the differential equations:

```
> eq4a1:=diff(x(t),t)=-5*x(t)+17*y(t);
```

$$\text{eq4a1} := \frac{d}{dt}x(t) = -5x(t) + 17y(t)$$

```
> eq4a2:=diff(y(t),t)=-2*x(t)^3+5*y(t);
```

$$\text{eq4a2} := \frac{d}{dt}y(t) = -2x(t)^3 + 5y(t)$$

Then, ask Maple to find a numerical solution to the initial-value problem:

```
> ans4a:=dsolve({eq4a1,eq4a2,x(0)=1,y(0)=-3},[x(t),y(t)],numeric);
```

```
ans4a := proc(x_rkf45) ... end proc;
```


To obtain the required function values at $t = 1$, we type

```
> ans4a(1);
```

$$[t = 1., x(t) = 3.41598087572498920, y(t) = -.950660885131171862]$$


To plot $x(t)$ and $y(t)$ versus t on the same graph, we use the command (having loaded the package `plots` already)

```
> odeplot(ans4a,[[t,x(t)],[t,y(t)]],0..2);
```



To create a parametric plot of $y(t)$ versus $x(t)$, use the command

```
> odeplot(ans4a, [x(t), y(t)], 0..2);
```



Note that the corners that appear on these graphs are not features of the true solutions, but are numerical artifacts. This concludes the solution of part 4a.

(b) $x' = x^2 - xy^2$, $y' = -y + x^3$, subject to $x(0) = 1$, $y(0) = -1$.

Note: your lab report for Lab 4 should consist of your solutions to the prelab problems, your solutions to lab problems 1–4 (including any graphs printed out), and a brief, original, conclusions paragraph summarizing what you have learned.